
Overview

■ Overview of the Output Delivery System (ODS)

■ Using the ODS EXCEL Destination

■ Customizing ODS EXCEL Output with Styles

Before the SAS Output Delivery System (ODS)

■ Traditional SAS “listing”

■ Designed for line printers

■ Monospace characters

■ No colors, fonts, or graphics

■ Difficult to import into other 

applications

Introducing the Output Delivery System (ODS)

■ ODS provides extensive flexibility in customizing SAS output

– Choose from numerous output file formats

– Utilize colors, fonts, borders, shading, etc.

– Embed graphics, images, links, tables, and more

– Control the arrangement and pagination of output

– Create accessible output that words with assistive technology

■ ODS is part of the Base SAS product since version 7
(No separate license required)

Components of the Output Delivery System

■ ODS destinations – control where the output goes

■ ODS objects – control how the output is stored and arranged

■ ODS styles – control how the output looks

■ ODS Statistical Graphics package

– An extension for creating analytical graphs

– Added in SAS 9.2

– A replacement for the aging SAS/GRAPH

From SAS Output Delivery System: User’s Guide, Fifth Edition, p. 34

USING THE ODS EXCEL 
DESTINATION

Section 1

The ODS EXCEL Destination

■ The ODS EXCEL destination creates native XLSX file output

■ Example syntax to write a PROC PRINT to an EXCEL file:

 ods excel file="c:\example.xlsx"; 

    <PROC PRINT code goes here...>;

 ods excel close;

ods excel file="c:\example.xlsx"; 

  proc print data=sashelp.cars; run;

ods excel close;

The ODS EXCEL OPTIONS option

■ The OPTIONS option provides various Excel-specific sub-options:

 ods excel file="c:\example.xlsx"

    options(subopt1=value1 subopt2=value2); 

    <PROC PRINT code goes here...>;

 ods excel close;

The ODS EXCEL OPTIONS option

■ Sub-options on the ODS EXCEL OPTIONS option can control:

– Sheet name

– Location on sheet for output

– Column width and row height

– Frozen row and column headers

– Hidden columns and rows

– Cell borders

– Excel Formulas

– More – refer to SAS 9.4 Output Delivery System User’s Guide

Specifying the Worksheet Name

■ The SHEET_NAME sub-option on the OPTIONS option of the ODS 

EXCEL statement can be used to specify the name of the Excel 

sheet (tab).

 ods excel file="c:\example.xlsx"

       options(sheet_name="My Output"); 

      <PROC PRINT code goes here...>;

 ods excel close;

ods excel file="c:\example.xlsx"

    options(sheet_name="My CARS Output"); 

proc print data=sashelp.cars;

run;

ods excel close;

Creating Multiple Tabs with BY Groups

■ A separate tab is added for each value of a BY variable.

■ Control the naming of tabs using the #BYVALn syntax

 ods excel file="c:\example.xlsx"

       options(sheet_name="#byval1"); 

      <PROC PRINT code goes here...>;

 ods excel close;

■ Remove the header on each page with the NOBYLINE option

 options nobyline;

proc sort data=sashelp.cars out=cars;

  by type make model;

run;

ods excel file="c:\example.xlsx"

        options(sheet_name="#byval1"); 

proc print data=cars;

  by type;

run;

ods excel close;

Specifying a Start Location

■ The START_AT sub-option on the OPTIONS option can be used to 

start a report somewhere other than Cell A1.

 ods excel file="c:\example.xlsx"

       options(start_at="C3"); 

      <PROC PRINT code goes here...>;

 ods excel close;

ods excel file="c:\example.xlsx"

    options(start_at="C3"); 

proc print data=sashelp.cars;

run;

ods excel close;

Adjusting Column Widths

■ By default, ODS sets column widths based on length of values/labels

■ Three ways to manually set column widths:

1. Override a specific column in PROC REPORT (or PRINT) using in-line 
styling on the DEFINE (or VAR) statement

2. Override all columns in PROC REPORT (or PRINT) using in-line styling on 
the PROC REPORT (or PROC PRINT) statement.

3. Override all columns using the ABSOLUTE_COLUMN_WIDTH suboption 
in the OPTIONS option of the ODS EXCEL statement.

 ods excel file="c:\example.xlsx"

       options(absolute_column_width="3in"); 

■ Various units available: in, cm, mm, px, pt, etc. Default = characters.

ods excel file="c:\example.xlsx"

    options(absolute_column_width="3in"); 

proc print data=sashelp.cars;

run;

ods excel close;

Adjusting Row Heights

■ Override all rows using the ABSOLUTE_ROW_HEIGHT sub-option in the 

OPTIONS option.

 ods excel file="c:\example.xlsx"

       options(absolute_row_height="30");

■ Specify different heights for header and body rows using the 

ROW_HEIGHTS sub-option in the OPTIONS option.

 ods excel file="c:\example.xlsx"

       options(row_heights="30,15");

■ Various units available: in, cm, mm, px, pt, etc. Default = pt (points).

ods excel file="c:\example.xlsx"

    options(absolute_row_height="30"); 

proc print data=sashelp.cars;

run;

ods excel close;

ods excel file="c:\example.xlsx"

    options(row_heights="30,15"); 

proc print data=sashelp.cars;

run;

ods excel close;

Hiding or Freezing Rows and Columns

■ The OPTIONS option provides sub-options to hide or freeze 
specific rows or columns:

 ods excel file="c:\example.xlsx"

  options(

   frozen_headers="1"

   frozen_rowheaders="2"

   hidden_rows="5,7"

   hidden_columns="6,8"

  );

To freeze columns or rows, 

specify the number of 

columns or rows to be frozen.

To hide columns or rows, 

specify the specific column or 

row numbers to be hidden.

ods excel file="c:\example.xlsx"

    options(

 frozen_headers="1"

 frozen_rowheaders="2"

 );

proc print data=sashelp.cars;

run;

ods excel close;

ods excel file="c:\example.xlsx"

    options(

 hidden_rows="5,7"

 hidden_columns="6,8"

 );

proc print data=sashelp.cars;

run;

ods excel close;

Removing Cell Borders

■ Cell borders can be removed by changing the template:

ods path(prepend) work.templat(update);

proc template;

  define style styles.mystyle;

    parent=styles.excel;

    class data, header, table, dataemphasis, linecontent /

      borderstyle=hidden;

  end;

run;

ods excel file="c:\example.xlsx" style=styles.mystyle;

ods excel file="c:\example.xlsx"

    style=styles.mystyle;

proc print data=sashelp.cars;

run;

ods excel close;

Modifying Cell Borders

■ Cell borders can also be modified by changing the template:

ods path(prepend) work.templat(update);

proc template;

  define style styles.mystyle;

    parent=styles.excel;

    class data, header, table, dataemphasis, linecontent /

      borderstyle=solid

      borderwidth=3px

      bordercolor=red;

  end;

run;

27

28



Creating and Customizing High-Impact Excel 
Workbooks from SAS with ODS EXCEL

Michigan SAS Users Group Conference
9/28/2023

Joshua M. Horstman
Copyright 2023 Nested Loop Consulting LLC 15

ods excel file="c:\example.xlsx"

    style=styles.mystyle;

proc print data=sashelp.cars;

run;

ods excel close;

Creating Excel Formulas from SAS

■ The FORMULAS suboption on the OPTIONS option allows you to 

specify if data values starting with an equal sign become Excel 

formulas or cell values:

 ods excel file="c:\example.xlsx"

       options(formulas="YES");

■ Build the desired Excel formulas in SAS and store them in SAS 

character variables before generating the Excel output.  

data cars;

  set sashelp.cars;

pricediff = cats("=C",_n_+1,"-D",_n_+1);

run;

ods excel file="c:\example.xlsx" options(formulas="YES");

proc print data=cars noobs label;

  label pricediff="Price Difference";

  var make model msrp invoice ;

  var pricediff / style={just=r tagattr='format:$#,##0_)'};

run;

ods excel close;

Output from Multiple Procedures

■ By default, output from each procedure is written to a separate 

tab within the Excel workbook. We can change the attributes of 

each tab independently:

ods excel file="c:\msug\example.xlsx" 

       options(tab_color="red" sheet_name="Cars"); 

proc print data=sashelp.cars; run;

ods excel options(tab_color="green" sheet_name="Class");

proc print data=sashelp.class; run;

ods excel close;

ods excel file="c:\msug\example.xlsx" 

       options(tab_color="red" sheet_name="Cars"); 

proc print data=sashelp.cars; run;

ods excel options(tab_color="green" sheet_name="Class");

proc print data=sashelp.class; run;

ods excel close;

Working with Dates

■ SAS and Excel represent dates and times differently

■ Unformatted SAS dates and times will not output correctly.

■ ODS EXCEL will correctly convert dates and times which have 

SAS date and time formats applied.

Dates Times Datetimes

SAS Days since 1/1/1960 Seconds since midnight Seconds since midnight 

1/1/1960

Excel Days since 1/1/1900 Fraction of day since 

midnight

Days (including fraction) 

since midnight 1/1/1900

ods excel file="c:\msug\example.xlsx"; 

proc print data=sashelp.stocks;

   format date;

run;

ods excel close;

ods excel file="c:\msug\example.xlsx"; 

proc print data=sashelp.stocks;

   format date date9.;

run;

ods excel close;

CUSTOMIZING ODS 
OUTPUT

Section 2

Using Built-in Styles

■ The appearance of ODS output can be modified using a style.

■ SAS includes a library of built-in styles:

– STYLES.BANKER

– STYLES.BRICK

– STYLES.HARVEST

– STYLES.OCEAN

– Many more…

■ Apply a style using the STYLE= option when opening a file:

ods excel file="c:\example.xlsx" style=styles.brown;

ods excel file="c:\example.xlsx"

style=styles.brown;

proc print data=sashelp.cars;

run;

ods excel close;

Creating Your Own Styles

■ You can create your own style using PROC TEMPLATE.

■ Your style can inherit from an existing one if desired.

 proc template;

  define style styles.mystyle;

   parent = styles.pearl;

   class data / color=red;

  end;

 run;

 ods excel file="c:\example.xlsx" style=styles.mystyle;

Available classes, elements, and attributes 

are extensive.  Refer to documentation.

ods excel file="c:\example.xlsx"

style=styles.mystyle;

proc print data=sashelp.cars;

run;

ods excel close;

Using In-Line Styling

■ Override appearance of individual elements using in-line styling:

  style(location)=[attribute=value …]

■ You can include one or more attributes and values.

■ Available attributes vary by system and destination – refer to 

ODS documentation.

■ Location is optional and is used in PROC REPORT and PROC 

TABULATE.

Using In-Line Styling in PROC REPORT

■ STYLE overrides are available for several report locations:

– COLUMN, HEADER, SUMMARY, REPORT, LINES, CALLDEF

■ Several PROC REPORT statements allow STYLE overrides:

Statement Default Location

PROC REPORT REPORT

DEFINE COLUMN and HEADER

BREAK / RBREAK SUMMARY and LINES

COMPUTE LINES

Using In-Line Styling in PROC REPORT

■ Example of customizing report output with in-line styling:

  ods excel file="c:\example.xlsx";

  proc report data=sashelp.cars style(report)=[cellpadding=3px];

    define make  / style(column)=[color=red];

    define model / style(header)=[color=green font_size=14pt];

    define type  / style=[background_color=blue color=white];

  run;

  ods excel close;

Additional Resources

45

46



Creating and Customizing High-Impact Excel 
Workbooks from SAS with ODS EXCEL

Michigan SAS Users Group Conference
9/28/2023

Joshua M. Horstman
Copyright 2023 Nested Loop Consulting LLC 24

Contact Information

Joshua M. Horstman

Nested Loop Consulting

josh@nestedloopconsulting.com

